

HT45R36 使用普通 I/O 驅動蜂鳴器


文件編碼：HA0119T

簡介

有些 MCU 由於沒有 PFD 功能，但在實際應用中需要蜂鳴器發聲，所以本範例以 HT45R36 為例，用普通 IO 模擬 PFD 功能，驅動 2K 的蜂鳴器。

應用電路圖

將蜂鳴器直接於 IO (PA3) 和地之間。通過控制 IO 的輸入輸出來控制 BZ 的輸出。



應用電路圖

軟體流程說明

```

.listinclude
INCLUDE HT45R36.INC
; ;*****
DATA_MAIN .SECTION 'DATA'           ;定義數據暫存器
R_ACC_BAK      DB ?                 ;累加器數據備份
R_STATUS_BAK   DB ?                 ;狀態暫存器數據備份
R_COUNTER1     DB ?                 ;定時計數器 1
R_COUNTER2     DB ?                 ;定時計數器 2
F_HALFSECOND   DBIT                ;0.5s 狀態旗標
;-----
CODE_MAIN .SECTION AT 00H 'CODE'     ;程式段
        JMP MLP_MAIN
        ORG 04H                      ;中斷 0 入口位址
        RETI
        ORG 08H                      ;中斷 1 入口位址
        RETI
        ORG 0CH                      ;定時/計數中斷入口位址
        JMP ISR_TIMER                ;跳到定時/計數中斷程式
        ORG 10H                      ;RC 振盪中斷入口位址
        RETI

ISR_TIMER:                               ;定時/計數中斷程式
MOV     R_ACC_BAK,A                     ;備份累加器資料
MOV     A,STATUS
MOV     R_STATUS_BAK,A                 ;備份狀態暫存器數據
MOV     A,08H
XORM    A,PA                           ;PA3 取反輸出
INC     R_COUNTER1                     ;進中斷一次定時計數器 1 加一
MOV     A,R_COUNTER1
XOR     A,80
SNZ     Z                               ;判斷是否計數到 (80*125us) =10ms
JMP     L_EXIT_ISR_TIMER               ;沒有計數到 10ms，退出中斷
CLR     R_COUNTER1                     ;已計數到，清除定時計數器 1 以便下次計數
INC     R_COUNTER2                     ;每計數到 10ms 時計數器 2 加一
MOV     A,R_COUNTER2
XOR     A,50
SNZ     Z                               ;判斷計數器 2 是否計數到 (50*10ms) =0.5s
JMP     L_EXIT_ISR_TIMER               ;沒有計數到 0.5s，退出中斷
CLR     R_COUNTER1                     ;已計數到，清除定時計數器 1 以便下次計數
CLR     R_COUNTER2                     ;已計數到，清除定時計數器 2 以便下次計數
SZ      F_HALFSECOND                  ;判斷 0.5s 狀態旗標，並取反
JMP     L_CLRFLAG                      ;0.5s 狀態旗標已置起，跳到清 0.5s 狀態旗標
SET     F_HALFSECOND                   ;置起 0.5s 狀態旗標
JMP     L_EXIT_ISR_TIMER               ;退出中斷

L_CLRFLAG:
CLR     F_HALFSECOND                   ;清除 0.5s 狀態旗標

L_EXIT_ISR_TIMER:
MOV     A,R_STATUS_BAK                 ;還原狀態暫存器數據
MOV     STATUS,A
MOV     A,R_ACC_BAK                   ;還原累加器數據
    
```

```

RETI
MLP_MAIN:
MOV     A, 09H           ;
MOV     INTC0, A        ;打開定時/計數器中斷
CLR     F_HALFSECOND   ;初始化 0.5s 狀態旗標，
CLR     R_COUNTER1     ;初始化定時計數器 1
CLR     R_COUNTER2     ;初始化定時計數器 2
SET     PAC.3           ;設置 PA3 為輸入，關閉 PFD
MOV     A, 06H
MOV     TMR, A          ;設置計數初始為 6，計數 250 次進一次中斷
MOV     A, 91H
MOV     TMRC, A         ;設置為內部定時，分頻 2MHz/2=1MHz，
                        ;250us 進一次中斷

LOOP:
SZ      F_HALFSECOND   ;判斷 0.5s 狀態旗標是置起
JMP     L_PFDOPEN      ;是，跳到打開 PFD

L_PFDCLOSE:
SET     PAC.3           ;否，設置 PA3 為輸入關閉 PFD
JMP     LOOP           ;循環判斷是否輸出 PFD

L_PFDOPEN:
CLR     PAC.3           ;是，設置 PA3 為輸出打開 PFD
JMP     LOOP           ;循環判斷是否輸出 PFD
    
```

軟體說明

這段程式的功能是蜂鳴器每隔 0.5s 叫一次，叫聲持續 0.5s。

程式通過設置定時 250us 進一次中斷，當 PA3 為輸出時，PA3 進中斷產生 2K 的方波以便驅動蜂鳴器發聲。

當設置 PA3 為輸入時，則 PA3 無法驅動蜂鳴器，蜂鳴器停止發聲。

程式中通過定時計數器 R_COUNTER1, R_COUNTER2 來定時時間 0.5s, 0.5s 到就取反 0.5s 狀態旗標，通過其旗標來循環開關蜂鳴器。